// Copyright (c) 2011, Jerzy Chalupski
// All rights reserved.
//
// Redistribution and use in source and binary forms, with or without
// modification, are permitted provided that the following conditions are met:
//     * Redistributions of source code must retain the above copyright
//       notice, this list of conditions and the following disclaimer.
//     * Redistributions in binary form must reproduce the above copyright
//       notice, this list of conditions and the following disclaimer in the
//       documentation and/or other materials provided with the distribution.
//     * Neither the name of the Jerzy Chalupski nor the
//       names of its contributors may be used to endorse or promote products
//       derived from this software without specific prior written permission.
//
// THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS" AND
// ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE IMPLIED
// WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE ARE
// DISCLAIMED. IN NO EVENT SHALL JERZY CHALUPSKI BE LIABLE FOR ANY
// DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR CONSEQUENTIAL DAMAGES
// (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF SUBSTITUTE GOODS OR SERVICES;
// LOSS OF USE, DATA, OR PROFITS; OR BUSINESS INTERRUPTION) HOWEVER CAUSED AND
// ON ANY THEORY OF LIABILITY, WHETHER IN CONTRACT, STRICT LIABILITY, OR TORT
// (INCLUDING NEGLIGENCE OR OTHERWISE) ARISING IN ANY WAY OUT OF THE USE OF THIS
// SOFTWARE, EVEN IF ADVISED OF THE POSSIBILITY OF SUCH DAMAGE.

import Qt 4.7

Item {
    id: runningChicken
    property int chickenState: -1
    property bool chickenLit
    property bool eggLit

    height: childrenRect.height
    width: childrenRect.width

    // ------------------------------------------------------------------------
    //  visible items
    // ------------------------------------------------------------------------
    //

    Row {
        id: chickenRow
        spacing: 5
        Repeater {
            model: [3, 2, 1]
            OldschoolLcd {
                imageSource: "graphics/chicken" + modelData + ".png"
                lit: runningChicken.chickenLit && chickenState == modelData
            }
        }
    }

    OldschoolLcd {
        id: hatchedChicken
        imageSource: "graphics/chicken0.png"
        lit: runningChicken.chickenLit && chickenState == 0
        anchors.left: chickenRow.right
        anchors.leftMargin: 5
        anchors.verticalCenter: chickenRow.top
    }

    Overlay {
        id: grass
        imageSource: "graphics/chickenbushes.png"
        anchors.verticalCenter: chickenRow.bottom
    }

    OldschoolLcd {
        id: brokenEgg
        imageSource: "graphics/brokenegg.png"

        anchors.left: hatchedChicken.right
        anchors.verticalCenter: grass.verticalCenter
        anchors.verticalCenterOffset: -5

        lit: eggLit
        onLitChanged: if (lit) breakEggAnimation.start();
    }

    // ------------------------------------------------------------------------
    //  animations
    // ------------------------------------------------------------------------
    //

    SequentialAnimation {
        id: breakEggAnimation
        PauseAnimation { duration: 20 * tick }
        ScriptAction { script: if (!chickenLit) lifeLossAnimationFinished() }
    }

    states: [
        State {
            name: "running"
            when: chickenLit
        }
    ]

    transitions: Transition {
         to: "running"
         SequentialAnimation {
             PauseAnimation { duration: 20 * tick }

             PropertyAction { target: runningChicken; property: "chickenState"; value: 0 }
             PauseAnimation { duration: 14 * tick }
             PropertyAction { target: runningChicken; property: "chickenState"; value: -1 }
             PauseAnimation { duration: tick }

             PropertyAction { target: runningChicken; property: "chickenState"; value: 1 }
             PauseAnimation { duration: 14 * tick }
             PropertyAction { target: runningChicken; property: "chickenState"; value: -1 }
             PauseAnimation { duration: tick }

             PropertyAction { target: runningChicken; property: "chickenState"; value: 2 }
             PauseAnimation { duration: 14 * tick }
             PropertyAction { target: runningChicken; property: "chickenState"; value: -1 }
             PauseAnimation { duration: tick }

             PropertyAction { target: runningChicken; property: "chickenState"; value: 3 }
             PauseAnimation { duration: 14 * tick }
             PropertyAction { target: runningChicken; property: "chickenState"; value: -1 }
             PauseAnimation { duration: tick }

             ScriptAction { script: lifeLossAnimationFinished() }
         }
    }
}
